

实验 6 报告

类MIPS多周期流水线处理器的设计与实现

李子龙

518070910095

2021 年 6 月 9 日

目录

1 实验目的	2
2 原理实现	2
2.1 基础流水线	2
2.2 前向转发机制	3
2.3 停顿机制	6
2.4 预测不发生机制	8
2.5 代码风格与顶端设计	9
3 仿真结果	15
3.1 思考问题	15
3.2 Mars 汇编程序	15
3.3 无优化仿真	16
3.4 前向传递	18
3.5 停顿机制	19
3.6 预测不发生机制	20
4 实验心得	20

1 实验目的

1. 理解CPU Pipeline，了解流水线冒险(hazard)及相关性，设计基础流水线CPU
2. 增加Forwarding机制解决数据竞争，减少因数据竞争带来的流水线停顿延时，提高流水线处理器性能
3. 设计支持Stall的流水线CPU。通过检测竞争并插入停顿（Stall）机制解决数据冒险、控制竞争和结构冒险
4. 通过predict-not-taken或延时转移策略解决控制冒险/竞争，减少控制竞争带来的流水线停顿延时，进一步提高处理器性能
5. 将CPU支持的指令数量从16条扩充为31条，使处理器功能更加丰富（选做）

2 原理实现

2.1 基础流水线




FIGURE 4.51 The pipelined datapath of Figure 4.46, with the control signals connected to the control portions of the pipeline registers. The control values for the last three stages are created during the instruction decode stage and then placed in the ID/EX pipeline register. The control lines for each pipe stage are used, and remaining control lines are then passed to the next pipeline stage.

通过添加段寄存器，以实现每个阶段的单独运行，并实现流水化。该图是基于 9 条指令的基础流水线设计。基于 16 条指令可以依据实验五进行改造。

2.2 前向转发机制

为了消除数据相关性（比如上一条指令的结果将会作为这一条指令运行），处理器需要引入前向转发机制（Forwarding）。




FIGURE 4.56 The datapath modified to resolve hazards via forwarding. Compared with the datapath in Figure 4.51, the additions are the multiplexers to the inputs to the ALU. This figure is a more stylized drawing, however, leaving out details from the full datapath, such as the branch hardware and the sign extension hardware.

表 1: 前向转发机制的控制信号

多选器控制	源	解释
ForwardA=00	ID	第一个 ALU 操作数来自寄存器堆
ForwardA=10	EX	第一个 ALU 操作数由上一个 ALU 运算结果转发获得
ForwardA=01	MEM	第一个 ALU 操作数从数据存储器或者前面的 ALU 运算结果中转发获得
ForwardB=00	ID	第二个 ALU 操作数来自寄存器堆
ForwardB=10	EX	第二个 ALU 操作数由上一个 ALU 运算结果转发获得
ForwardB=01	MEM	第二个 ALU 操作数由数据存储器或者前面的 ALU 结果转发获得

对于两种 EX 冒险和 MEM 冒险两种模式，根据表 1 采用以下的判定方法以及输出方式。

```

1  wire [1:0] ForwardA = ((EX_REG_WRITE &
2      (EX_WRITE_REG != 0) &
3      (EX_WRITE_REG == ID_INST[25:21])) ?
4      2'b10 :
5      ((MEM_REG_WRITE & (MEM_WRITE_REG != 0)
6      & (MEM_WRITE_REG == ID_INST[25:21])) ?
7      2'b01 :
8      2'b00));
9
10 wire [1:0] ForwardB = ((EX_REG_WRITE &
11     (EX_WRITE_REG != 0) &
12     (EX_WRITE_REG == ID_INST[20:16])) ?
13     2'b10 :
14     ((MEM_REG_WRITE & (MEM_WRITE_REG != 0)
15     & (MEM_WRITE_REG == ID_INST[20:16])) ?
16     2'b01 :
17     2'b00));
18
19 wire [31:0] ForwardA_RES =
20     ForwardA == 2'b00 ? ID_READ_DATA1 :
21     (ForwardA == 2'b10 ? EX_ALU_RES :
22     WRITE_DATA_WB);
23
24 wire [31:0] ForwardB_RES =
25     ForwardB == 2'b00 ? ID_READ_DATA2 :
26     (ForwardB == 2'b10 ? EX_ALU_RES :
27     WRITE_DATA_WB);

```

这样，EX 冒险（10）就可以由 ALU 转发，而 MEM 冒险（01）就可以由写回数据转发。而对于接收者也要做一定的适配，并且为了匹配立即数指令，需要在转发结果后再加选路器。




FIGURE 4.57 A close-up of the datapath in [Figure 4.54](#) shows a 2:1 multiplexor, which has been added to select the signed immediate as an ALU input.

```

1      ALU alu(
2          .input1(SHAMT ? ID_INST[10:6] : ForwardA_RES), // ForwardA
3          .input2(ID_ALU_SRC ? ID_OPRAND : ForwardB_RES), // ForwardB
4          .aluCtr(ALU_CTR),
5          .zero(ZERO_EX),
6          .aluRes(ALU_RES_EX)
7      );
8
9      dataMemory DataMemory(
10         .Clk(clk),
11         .address(EX_ALU_RES),
12         .writeData(ForwardB_RES), // ForwardB
13         .memWrite(EX_MEM_WRITE),
14         .memRead(EX_MEM_READ),
15         .readData(READ_DATA_MEM)
16     );

```

值得注意的是，存储器的写入数据使用了前向传递B的结果。

并且在 WB 级不会发生冒险的前提是

假设在 ID 级指令读取的存储器与 WB 级指令写入的寄存器是同一个寄存器时，就由寄存器堆提供正确的结果。

因此，寄存器堆的输出不能依赖于输入地址的变化，一旦数据出现了变化，就需要立刻输出更新值。这可以采用 `wire` 类型直接赋值实现，这样一旦时钟下沿写入，就会被立刻更新为新值，从而影响下一时钟上沿的赋值。

Listing 1: `Registers.v`

```

1 module Registers(
2     input [25:21] readReg1,
3     input [20:16] readReg2,
4     input [4:0] writeReg,
5     input [31:0] writeData,
6     input regWrite,
7     output [31:0] readData1,
8     output [31:0] readData2,
9     input clk,
10    input reset
11 );
12
13    reg [31:0] RegFile [31:0];
14
15    // reg [31:0] ReadData1;
16    // reg [31:0] ReadData2;
17    // always @(readReg1 or readReg2) begin
18    //     ReadData1 = RegFile[readReg1];
19    //     ReadData2 = RegFile[readReg2];
20    // end
21    // assign readData1 = ReadData1;
22    // assign readData2 = ReadData2;
23    assign readData1 = RegFile[readReg1];
24    assign readData2 = RegFile[readReg2];
25
26    integer i;
27    always @(negedge clk or reset) begin
28        if(reset) begin
29            for(i = 0; i < 32; i = i + 1)
30                RegFile[i] = 0;
31        end
32        else begin
33            if(regWrite)
34                RegFile[writeReg] = writeData;
35        end
36    end
37 endmodule

```

2.3 停顿机制

当寄存器尚未获得值且紧随的运算需要该寄存器的后值时，仅仅前向转发是不足的。这时候就要引入停顿机制。




FIGURE 4.60 Pipelined control overview, showing the two multiplexers for forwarding, the hazard detection unit, and the forwarding unit. Although the ID and EX stages have been simplified—the sign-extended immediate and branch logic are missing—this drawing gives the essence of the forwarding hardware requirements.

用来解决 lw - use 冒险，就需要考察读取的存储器是不是使用的存储器。

```

wire stalling = (ID_MEM_READ &&
  ((ID_INST[20:16] == IF_INST[25:21])
  || (ID_INST[20:16] == IF_INST[20:16]))) ?
  1 : 0;

```

一旦停顿机制被激活，就会向流水线中添加一个气泡以停顿，方法就是IF 值被维持和ID 控制信号清除（这一步预测不发生共用）。

```

1  if (stalling || (ID_BRANCH && BEQ)) begin
2    // nop
3    ID_REG_DST <= 0;
4    ID_JUMP <= 0;
5    ID_BRANCH <= 0;
6    ID_MEM_READ <= 0;
7    ID_MEM_TO_REG <= 0;
8    ID_MEM_WRITE <= 0;
9    ID_ALU_OP <= 0;
10   ID_ALU_SRC <= 0;

```

```

11     ID_IMM <= 0;
12     ID_JAL <= 0;
13     ID_REG_WRITE <= 0;
14     else //...
15
16     // IF
17     if(stalling == 0) begin
18         PC <= PC + 4;
19         IF_PC <= PC;           // Has already been PC + 4
20         IF_INST <= INST_IF;
21     end // stalling means no change

```

2.4 预测不发生机制

最后来处理分支情况。如果采用传统的方式，发生分支时，需要等待分支在 EX 段才可以判断，此前需要一直阻塞（3个周期）。这样做实在太慢，不利于流水线的运转。




FIGURE 4.62 The ID stage of clock cycle 3 determines that a branch must be taken, so it selects 72 as the next PC address and zeros the instruction fetched for the next clock cycle. Clock cycle 4 shows the instruction at location 72 being fetched and the single bubble or nop instruction in the pipeline as a result of the taken branch. (Since the nop is really `sl $0, $0, 0`, it's arguable whether or not the ID stage in clock 4 should be highlighted.)

仔细阅读课本后，将做如下的设计：遇到分支认为不运行分支，计数器继续递增。而将 branch 判定提前到 ID 段，一旦发现判断错误，就会向流水线中加入一个气泡，并将 PC 设定为分支位置，将 IF 寄存器清空（IF.Flush）。气泡是在时钟上沿进行（前文已提到），而

后面的需要在时钟下沿进行，为了影响后面所有指令的进行（在后文的处理器顶端模块编写中可以看到赋值顺序是从后向前进行的，以传递数据，这样做可以在不影响整个代码风格的情况下，以影响后面的指令，否则只会不会在前面阻塞，后面赋值又会覆盖）。

```

1  always @(negedge clk ) begin
2      // Here is the clear signal before transition.
3      // IF.Flush
4      if(ID_BRANCH && BEQ) begin
5          PC <= BRANCH_PC_ID;
6          IF_PC <= 0;
7          IF_INST <= 0;
8      end
9      //...
10 end

```

判断的方法需要直接使用前向传递的数据，判定两个操作数是否相等。否则会在特定的事件产生判断错误的结果（比如分支指令前恰好是 ALU 指令，这时必然要阻塞）。

```

1  // predict-not-taken
2  wire BEQ = (ForwardA_RES == ForwardB_RES);

```

2.5 代码风格与顶端设计

有了上面的考虑之后，就可以进行流水线的顶端设计了。

本处理器的代码风格如下：

- 所有的段寄存器都会采用前一阶段的名称作为开始。比如 IF/ID 寄存器中的 INST，就会被定义为

```

1  reg [31:0] IF_INST;

```

以表示该值由 IF 阶段发出。

- 所有的电线采用终止寄存器结尾。比如 IF 阶段的指令存储器发出指令信息以存储到 IF/ID 寄存器，就会被定义下面的方式，表示电线将会在 IF/ID 寄存器终止。

```

1  wire [31:0] INST_IF;

```

- 所有段寄存器的值会在时钟上跳沿被逆阶段更新，也就意味着寄存器的值将会晚电线值一个周期。比如

```
1      always @(posedge clk) begin
2          // reset
3          // WB
4          // MEM
5          // EX
6          // ID
7          ID_INST <= IF_INST;
8          // IF
9          IF_INST <= INST_IF;
10     end
```

这样做就可以防止不确定的赋值顺序。当需要实时值时需要使用以阶段结尾的电线值（一般在时钟沿之外使用），在时钟沿更新时如果没有对应的直接输出电线，就要采用寄存器值更新。

- 当需要影响全阶段的运行时，需要在时钟下沿更新值。这一点与实验五的设计是一致的。比如跳转（含 JR 指令）、分支指令判断错误等：

```
1      always @(negedge clk ) begin
2          // IF.Flush
3          if(ID_BRANCH && BEQ) begin
4              PC <= BRANCH_PC_ID;
5              IF_PC <= 0;
6              IF_INST <= 0;
7          end
8          if(EX_JR) begin
9              PC <= EX_ALU_RES;
10             IF_PC <= 0;
11             IF_INST <= 0;
12             ID_PC <= 0;
13             ID_INST <= 0;
14             ID_REG_DST <= 0;
15             ID_JUMP <= 0;
16             ID_BRANCH <= 0;
17             ID_MEM_READ <= 0;
18             ID_MEM_TO_REG <= 0;
19             ID_MEM_WRITE <= 0;
```

```

20     ID_ALU_OP <= 0;
21     ID_ALU_SRC <= 0;
22     ID_IMM <= 0;
23     ID_JAL <= 0;
24     ID_REG_WRITE <= 0;
25   end else
26     PC <= EX_JUMP ? EX_JUMP_PC : PC;
27   end

```




FIGURE 4.65 The final datapath and control for this chapter. Note that this is a stylized figure rather than a detailed datapath, so it's missing the ALUSrc Mux from Figure 4.57 and the multiplexor controls from Figure 4.51.

Listing 2: Top.v

```

1  `timescale 1ns / 1ps
2  //////////////////////////////////
3  // Company:
4  // Engineer: Zilong Li
5  //
6  // Create Date: 2021/06/05 12:37:39
7  // Design Name:
8  // Module Name: Top
9  // Project Name:
10 // Target Devices:
11 // Tool Versions:
12 // Description:
13 //
14 // Dependencies:
15 //
16 // Revision:
17 // Revision 0.01 - File Created
18 // Additional Comments:
19 //
20 //////////////////////////////////
21
22
23 module Top(
24     input clk,
25     input reset
26 );
27

```

```

28 // Instruction Fetch (IF)
29 reg [31:0] PC;
30 reg [31:0] IF_PC;
31
32 wire [31:0] INST_IF;
33
34 reg [31:0] IF_INST;
35
36 // Instruction Decoding (ID)
37 reg [31:0] ID_PC;
38
39 wire REG_DST_ID;
40 wire JUMP_ID;
41 wire BRANCH_ID;
42 wire MEM_READ_ID;
43 wire MEM_TO_REG_ID;
44 wire MEM_WRITE_ID;
45 wire [1:0] ALU_OP_ID;
46 wire ALU_SRC_ID;
47 wire IMM_ID;
48 wire JAL_ID;
49 wire REG_WRITE_ID;
50
51 wire ZEXT;
52
53 wire [31:0] READ_DATA1_ID;
54 wire [31:0] READ_DATA2_ID;
55
56 wire [31:0] OPRAND_ID;
57
58 reg ID_REG_DST;
59 reg ID_JUMP;
60 reg ID_BRANCH;
61 reg ID_MEM_READ;
62 reg ID_MEM_TO_REG;
63 reg ID_MEM_WRITE;
64 reg [1:0] ID_ALU_OP;
65 reg ID_ALU_SRC;
66 reg ID_IMM;
67 reg ID_JAL;
68 reg ID_REG_WRITE;
69
70 reg [31:0] ID_READ_DATA1;
71 reg [31:0] ID_READ_DATA2;
72 reg [31:0] ID_OPRAND;
73 reg [31:0] ID_INST;
74
75 wire [31:0] BRANCH_PC_ID = IF_PC + 4 + (OPRAND_ID <<
    2); // hasn't been PC + 4
76
77 // Execution (EX)
78 reg [31:0] EX_PC;
79 reg [31:0] EX_JUMP_PC;
80
81 wire ZERO_EX;
82 wire [31:0] ALU_RES_EX;
83 wire JR_EX;
84
85 wire [3:0] ALU_CTR;
86 wire SHAMT;
87
88 reg EX_JR;
89 reg EX_ZERO;
90 reg [31:0] EX_ALU_RES;
91 reg [4:0] EX_WRITE_REG;
92
93 reg EX_JUMP;
94 reg EX_BRANCH;
95 reg EX_MEM_READ;
96 reg EX_MEM_TO_REG;
97 reg EX_MEM_WRITE;
98 reg EX_JAL;
99 reg EX_REG_WRITE;
100
101 reg [31:0] EX_READ_DATA2;
102
103 // Memory (MEM)
104 reg [31:0] MEM_PC;
105
106 wire PC_SRC = EX_BRANCH & EX_ZERO;
107 wire [31:0] READ_DATA_MEM;
108
109 reg MEM_REG_WRITE;
110 reg MEM_MEM_TO_REG;
111 reg MEM_JAL;
112
113 reg [31:0] MEM_READ_DATA;
114 reg [31:0] MEM_ALU_RES;
115 reg [4:0] MEM_WRITE_REG;
116
117 // Write Back (WB)
118 wire REG_WRITE_WB = MEM_REG_WRITE;
119 wire [31:0] WRITE_DATA_WB = MEM_JAL ? MEM_PC + 4 : (
    MEM_MEM_TO_REG ? MEM_READ_DATA : MEM_ALU_RES);
    // The next PC for jal
120 wire [4:0] WRITE_REG_WB = MEM_WRITE_REG;
121
122 // Forwarding
123 wire [1:0] ForwardA = ((EX_REG_WRITE &
    (EX_WRITE_REG != 0) &
    (EX_WRITE_REG == ID_INST[25:21])) ?
    2'b10 :
    ((MEM_REG_WRITE & (MEM_WRITE_REG != 0)
    & (MEM_WRITE_REG == ID_INST[25:21])) ?
    2'b01 :
    2'b00));
124
125 wire [1:0] ForwardB = ((EX_REG_WRITE &
    (EX_WRITE_REG != 0) &
    (EX_WRITE_REG == ID_INST[20:16])) ?
    2'b10 :
    ((MEM_REG_WRITE & (MEM_WRITE_REG != 0)
    & (MEM_WRITE_REG == ID_INST[20:16])) ?
    2'b01 :
    2'b00));
126
127
128
129
130
131
132
133
134
135
136
137
138
139
140

```

```

141 wire [31:0] ForwardA_RES =
142     ForwardA == 2'b00 ? ID_READ_DATA1 :
143     (ForwardA == 2'b10 ? EX_ALU_RES :
144     WRITE_DATA_WB);
145
146 wire [31:0] ForwardB_RES =
147     ForwardB == 2'b00 ? ID_READ_DATA2 :
148     (ForwardB == 2'b10 ? EX_ALU_RES :
149     WRITE_DATA_WB);
150
151
152 // Stall
153 wire stalling = (ID_MEM_READ & ((ID_INST[20:16] ==
154     IF_INST[25:21]) | (ID_INST[20:16] == IF_INST
155     [20:16]))) ? 1 : 0;
156
157 // predict-not-taken
158 wire BEQ = (ForwardA_RES == ForwardB_RES);
159
160 // IF
161 InstMemory instMemory(
162     .readAddress(PC),
163     .inst(INST_IF)
164 );
165
166 // ID
167 Ctr mainCtr(
168     .opCode(IF_INST[31:26]),
169     .regDst(REG_DST_ID),
170     .jump(JUMP_ID),
171     .branch(BRANCH_ID),
172     .memRead(MEM_READ_ID),
173     .memToReg(MEM_TO_REG_ID),
174     .memWrite(MEM_WRITE_ID),
175     .aluOp(ALU_OP_ID),
176     .aluSrc(ALU_SRC_ID),
177     .imm(IMM_ID),
178     .regWrite(REG_WRITE_ID),
179     .jal(JAL_ID),
180     .zext(ZEXT)
181 );
182
183 Registers registers(
184     .clk(clk),
185     .reset(reset),
186     .readReg1(IF_INST[25:21]),
187     .readReg2(IF_INST[20:16]),
188     .writeReg(WRITE_REG_WB), // After write
189     .back, // ...
190     .writeData(WRITE_DATA_WB), // ...
191     .regWrite(REG_WRITE_WB), // ...
192     .readData1(READ_DATA1_ID),
193     .readData2(READ_DATA2_ID)
194 );
195
196 signext signExt(
197     .inst(IF_INST[15:0]),
198     .zext(ZEXT),
199
200     .data(OPRAND_ID)
201 );
202
203 // EX
204 ALUCtr aluCtr(
205     .nop(ID_INST == 0 ? 1'b1 : 1'b0),
206     .funct(ID_IMM ? ID_INST[31:26] : ID_INST[5:0]),
207     .aluOp(ID_ALU_OP),
208     .aluCtrOut(ALU_CTR),
209     .jrr(JR_EX),
210     .shamt(SHAMT)
211 );
212
213 ALU alu(
214     .input1(SHAMT ? ID_INST[10:6] : ForwardA_RES),
215     .input2(ID_ALU_SRC ? ID_OPRAND : ForwardB_RES),
216     .aluCtr(ALU_CTR),
217     .zero(ZERO_EX),
218     .aluRes(ALU_RES_EX)
219 );
220
221 // MEM
222 dataMemory DataMemory(
223     .Clk(clk),
224     .address(EX_ALU_RES),
225     .writeData(ForwardB_RES),
226     .memWrite(EX_MEM_WRITE),
227     .memRead(EX_MEM_READ),
228     .readData(READ_DATA_MEM)
229 );
230
231 always @(posedge clk) begin
232     if(reset) begin
233         PC <= 0;
234         IF_PC <= 0;
235         IF_INST <= 0;
236
237         ID_PC <= 0;
238         ID_READ_DATA1 <= 0;
239         ID_READ_DATA2 <= 0;
240         ID_OPRAND <= 0;
241         ID_INST <= 0;
242
243         ID_REG_DST <= 0;
244         ID_JUMP <= 0;
245         ID_BRANCH <= 0;
246         ID_MEM_READ <= 0;
247         ID_MEM_TO_REG <= 0;
248         ID_MEM_WRITE <= 0;
249         ID_ALU_OP <= 0;
250         ID_ALU_SRC <= 0;
251         ID_IMM <= 0;
252         ID_JAL <= 0;
253         ID_REG_WRITE <= 0;
254
255         EX_PC <= 0;
256         EX_ZERO <= 0;
257         EX_ALU_RES <= 0;

```

```

254     EX_WRITE_REG <= 0;
255
256     EX_JUMP <= 0;
257     EX_BRANCH <= 0;
258     EX_MEM_READ <= 0;
259     EX_MEM_TO_REG <= 0;
260     EX_MEM_WRITE <= 0;
261     EX_REG_WRITE <= 0;
262
263     EX_READ_DATA2 <= 0;
264
265     MEM_REG_WRITE <= 0;
266
267     MEM_READ_DATA <= 0;
268     MEM_ALU_RES <= 0;
269     MEM_WRITE_REG <= 0;
270
271     end
272     else begin
273         // WB
274
275         // MEM
276         MEM_PC <= EX_PC;
277
278         MEM_REG_WRITE <= EX_REG_WRITE;
279         MEM_MEM_TO_REG <= EX_MEM_TO_REG;
280
281         MEM_READ_DATA <= READ_DATA_MEM;
282         MEM_ALU_RES <= EX_ALU_RES;
283         MEM_WRITE_REG <= EX_WRITE_REG;
284         MEM_JAL <= EX_JAL;
285
286         // EX
287         EX_PC <= ID_PC;
288
289         EX_ZERO <= ZERO_EX;
290         EX_ALU_RES <= ALU_RES_EX;
291         EX_WRITE_REG <= ID_JAL ? 5'b11111 : (JR_EX ?
292             ID_INST[25:21] : (ID_REG_DST ? ID_INST
293                 [15:11] : ID_INST[20:16]));
294
295         EX_JUMP <= ID_JUMP;
296         EX_BRANCH <= ID_BRANCH;
297         EX_MEM_READ <= ID_MEM_READ;
298         EX_MEM_TO_REG <= ID_MEM_TO_REG;
299         EX_MEM_WRITE <= ID_MEM_WRITE;
300         EX_REG_WRITE <= ID_REG_WRITE;
301         EX_JAL <= ID_JAL;
302         EX_JR <= JR_EX;
303
304         EX_JUMP_PC <= ID_PC[31:28] + (ID_INST[25:0]
305             << 2);
306
307         EX_READ_DATA2 <= ID_READ_DATA2;
308
309         // ID
310         ID_PC <= IF_PC;
311         ID_INST <= IF_INST;
312         ID_READ_DATA1 <= READ_DATA1_ID;

```

```

309     ID_READ_DATA2 <= READ_DATA2_ID;
310     ID_OPRAND <= OPRAND_ID;
311
312     if (stalling || (ID_BRANCH && BEQ)) begin
313         // nop
314         ID_REG_DST <= 0;
315         ID_JUMP <= 0;
316         ID_BRANCH <= 0;
317         ID_MEM_READ <= 0;
318         ID_MEM_TO_REG <= 0;
319         ID_MEM_WRITE <= 0;
320         ID_ALU_OP <= 0;
321         ID_ALU_SRC <= 0;
322         ID_IMM <= 0;
323         ID_JAL <= 0;
324         ID_REG_WRITE <= 0;
325     end else begin
326         ID_REG_DST <= REG_DST_ID;
327         ID_JUMP <= JUMP_ID;
328         ID_BRANCH <= BRANCH_ID;
329         ID_MEM_READ <= MEM_READ_ID;
330         ID_MEM_TO_REG <= MEM_TO_REG_ID;
331         ID_MEM_WRITE <= MEM_WRITE_ID;
332         ID_ALU_OP <= ALU_OP_ID;
333         ID_ALU_SRC <= ALU_SRC_ID;
334         ID_IMM <= IMM_ID;
335         ID_JAL <= JAL_ID;
336         ID_REG_WRITE <= REG_WRITE_ID;
337     end
338
339     // IF
340     if(stalling == 0) begin
341         PC <= PC + 4;
342         IF_PC <= PC; // Has already been
343             PC + 4
344         IF_INST <= INST_IF;
345     end
346
347     end
348
349     always @(negedge clk) begin
350         // Here is the clear signal before transition.
351         // IF.Flush
352         if(ID_BRANCH && BEQ) begin
353             PC <= BRANCH_PC_ID;
354             IF_PC <= 0;
355             IF_INST <= 0;
356         end
357         if(EX_JR) begin
358             PC <= EX_ALU_RES;
359             IF_PC <= 0;
360             IF_INST <= 0;
361             ID_PC <= 0;
362             ID_INST <= 0;
363             ID_REG_DST <= 0;
364             ID_JUMP <= 0;
365             ID_BRANCH <= 0;
366             ID_MEM_READ <= 0;

```

```
366 ID_MEM_TO_REG <= 0;  
367 ID_MEM_WRITE <= 0;  
368 ID_ALU_OP <= 0;  
369 ID_ALU_SRC <= 0;  
370 ID_IMM <= 0;  
371 ID_JAL <= 0;  
372 ID_REG_WRITE <= 0;
```

```
373     end else  
374         PC <= EX_JUMP ? EX_JUMP_PC : PC;  
375     end  
376  
377 endmodule
```

3 仿真结果

3.1 思考问题

在完成实现后，先回答一些思考题目。

1. 请思考和Lab5相比，Top模块中的主要变化处是什么？

答 需要考虑赋值的顺序，不能像 Lab 5 一样将电线直接接上。更重要的是，需要考虑冒险问题。

2. 之前的模块是否要修改？

答 需要修改，比如 Registers 的相关修改，见 2.2。

3. 另外，由于MEM级的Branch会影响PCSrc的值，从而影响下次PC，因此需要为Control加入RESET功能，将Branch置零。

答 该问题在预测不发生机制统一考虑，见 2.4。

4. 由于各种变量名称极为复杂，推荐在着手编码之前为自己选择一套命名规范

答 见代码风格 2.5。


5. 在实现实验目的2.，3.，4.的内容时建议把1.的代码或工程备份一遍才开始

答 本实验采用 git 版本控制工具统一管理。

3.2 Mars 汇编程序


为了加快测试进度，在本实验中采用计算机结构理论课上的 Mars 工具快速生成汇编二进制代码。

打开 Mars。新建 asm 汇编程序，然后汇编后，在文件选单中 Dump Memory，另存为 Binary Text 就可以生成二进制的程序码。值得注意的是，需要采用标签指定 beq 的目标位置，也就是实验指导书中的示例汇编代码风格。




3.3 无优化仿真

首先采用教材代码，然后会有数据冒险问题。



手动插入空语句后（lw-use型需要三个空语句），就可以解决数据冒险问题。从而得到预期结果。



Listing 3: example.asm

```

1  lw $1, 0($0)
2  lw $2, 4($0)
3  lw $3, 8($0)
4  add $4, $1, $2
5  sub $5, $3, $1
6  and $6, $2, $1
7  lw $10, 0($0)
8  lw $10, 0($0)
9  lw $10, 0($0)
10 or $7, $3, $1
11 slt $8, $3, $1
12 beq $0, $0, end
13 add $9, $7, $8
14 end:lw $10, 0($0)
15 lw $10, 0($0)
16 lw $10, 0($0)
17 lw $10, 0($0)
18 lw $10, 0($0)

```

Listing 4: example.asm(no risk)

```

1  lw $1, 0($0)
2  lw $2, 4($0)
3  lw $3, 8($0)
4  nop
5  nop
6  nop
7  add $4, $1, $2
8  sub $5, $3, $1
9  and $6, $2, $1
10 lw $10, 0($0)
11 lw $10, 0($0)
12 lw $10, 0($0)
13 or $7, $3, $1
14 slt $8, $3, $1
15 beq $0, $0, end
16 add $9, $7, $8
17 end:lw $10, 0($0)
18 lw $10, 0($0)
19 lw $10, 0($0)
20 lw $10, 0($0)
21 lw $10, 0($0)

```

3.4 前向传递

从本小节起，采用于实验 5 同样的代码来说明问题。

该指令文件主要的作用是测试所有的运算功能，并在每一个循环对 10 号寄存器 + 1，并存储到 0 号存储单元中，直到其超过刚开始的限制寄存器的存储数字（这里是 4），之后就会进入短循环，不会再对寄存器和存储器进行修改。

Listing 5: `simple.asm`

```

1 nop
2 lw $16, 8($0)          # $0 zero register
3 jal 4
4 nop
5 lw $8, 0($0)
6 lw $9, 4($0)
7 sub $10, $8, $9
8 and $10, $8, $9


```

```


9 slt $10, $8, $9
10 or $10, $8, $9
11 addi $10, $8, 8
12 andi $10, $8, -1
13 ori $10, $8, -1
14 sll $10, $8, 1
15 srl $10, $8, 1
16 add $10, $8, $9        # final save: += 1
17 sw $10, 0($0)
18 beq $10, $16, 1
19 jr $31
20 j 16
21 nop
22 nop

```

首先仿真得到了与上个实验相同的结果，并达到了运行预期（4 之后不更改）。




对于前向传递，可以看到由于 ForwardA 被置于 1，ALU 第一个输入就会被导入写回数据；由于 ForwardB 被置于 2，ALU 第二个输入就会被导入上一个 ALU 的结果。结果如下图所示。该前向传递发生在开始的运算时期，此时的 lw 尚未写回寄存器，但是需要其值。




3.5 停顿机制

同一个时期，由于前向传递还不足，就需要停顿一个周期，这个时候 PC 就会被拉长。与此对应的是，所有 ID 控制信号会被清空。



3.6 预测不发生机制

当最后一个判断时，由于分支发生了，所以将会插入一个气泡，以清除，并且将PC的值进入短循环。



4 实验心得

计算机系统结构实验终于结束，奋战 4 日的流水处理器终于完成。在设计流水处理器的时候，一直在重新翻看课本，复习概念。在这个过程中加深了我对流水处理器的许多理解。

由于时间不足，所以 31 条指令的选做任务并没有完成，由于需要大改，可能会对代码风格产生影响，并且需要增加很多端口。做完一些减少停顿的策略，理解如何解决冒险足矣。

多次的失败，让我意识到是自己对细节的把握不足，认真看书，总是可以找到对应的答案。感谢实验组给我这个设计处理器的机会，该实验对于我理解处理器的核心细节有很大的帮助。

灵活使用课上的知识，可以更好的解决问题，提高效率（比如使用 Mars 汇编程序）。形成良好的代码风格有助于减少问题发生的几率。并且在无法调试出问题时，不妨建立最小工作集，新建波形，通过代码追踪相关端口，以解决问题。

本实验的处理器原理图形来自参考文献的课本。

参考文献

- [1] Computer Organization and Design — The Hardware/Software Interface, David A. Patterson and John L. Hennessy, Fifth Edition, Elsevier Inc., 2014